		
	use "$data/secondary/final_data.dta", clear
		
	******************************************************
	***		 TABLES S3 and S4 - BALANCE TEST   	 	   ***
	******************************************************	
		
	global controls "white black mixed other_race bweight bheight plannedbirth firstborn agebirth prenatal naturalbirth preterm breastfed hhld_income hhld_size caregiver_age caregiver_literacy caregiver_middleschool caregiver_highschool caregiver_college"
		
	forvalues male = 0 / 1 {
		
		foreach var in $controls {
		
			mean `var'		if ITT == 0 & male == `male'
				
			areg `var' ITT 	if male == `male', absorb(cluster) robust
				
		}
		
	}

	******************************************************
	***			TABLE S5 - FIRST STAGE		   	   	   ***
	******************************************************	
		
	xtset cluster
			
	forvalues i = 0/4 {
				
		gen depvar_`i' = (daycare_attendance == `i') if daycare_attendance < .
				
		mean depvar_`i' if ITT == 0 & male == 0
				
		mean depvar_`i' if ITT == 0 & male == 1
				
		xtreg depvar_`i' male_ITT female_ITT male, robust fe
			
	}

	*********************************************************
	***	FIG S5 - ITT on MEASURES of DECISION-MAKING QUALITY *
	*********************************************************
	
	******************************************************
	******************************************************
	
	cap program drop CDF_boys
	program define CDF_boys
	
			*1: dependent variable
			*2: title of x-axis
			*3: xscale
			*4: xlabel
			*5: far left tail of distribution of DMQ index not shown
			*6: title of graph
			
			cap drop Cntrl_Boys Treat_Boys
			
			cumul `1' if male== 1 & ITT == 0, gen(Cntrl_Boys) equal
			cumul `1' if male== 1 & ITT == 1, gen(Treat_Boys) equal

			`5'
			
			twoway 	(line Cntrl_Boys `1' if ITT == 0 & male == 1, sort lcolor(ebblue) 	 lwidth(med) lpattern(dash)  connect(stairstep)) ///	
					(line Treat_Boys `1' if ITT == 1 & male == 1, sort lcolor(eltblue) 	 lwidth(med) lpattern(solid)  connect(stairstep)) ///				
					, ylabel(, nogrid) graphregion(fcolor(white)) xtitle("`2'") ytitle("Cumulative Distribution") /// 
					title("`6'", size(medium) ring(0) position(12)) legend(order(1 "Control Boys" 2 "Treatment Boys") region(lstyle(none))) ///
					yscale(range(0 1)) ylabel(0(.2)1) ///
					xscale(range(`3')) xlabel(`4')
					
	end			
	
	cap program drop CDF_girls
	program define CDF_girls
	
			*1: dependent variable
			*2: title of x-axis
			*3: xscale
			*4: xlabel
			*5: far left tail of distribution of DMQ index not shown
			*6: title of graph
			
			cap drop Cntrl_Girls Treat_Girls
			
			cumul `1' if male== 0 & ITT == 0, gen(Cntrl_Girls) equal
			cumul `1' if male== 0 & ITT == 1, gen(Treat_Girls) equal
			
			`5'
			
			twoway 	(line Cntrl_Girls `1' if ITT == 0 & male == 0, sort lcolor(pink*1.5) 	 lwidth(med) lpattern(dash)  connect(stairstep)) ///	
					(line Treat_Girls `1' if ITT == 1 & male == 0, sort  lcolor(pink*0.5) 	 lwidth(med) lpattern(solid)  connect(stairstep)) ///				
					, ylabel(, nogrid) graphregion(fcolor(white)) xtitle("`2'") ytitle("Cumulative Distribution") /// 
					title("`6'", size(medium) ring(0) position(12)) legend(order(1 "Control Girls" 2 "Treatment Girls") region(lstyle(none))) /// 
					yscale(range(0 1)) ylabel(0(.2)1) ///
					xscale(range(`3')) xlabel(`4')
			
	end				
	
	********************************************************
	********************************************************
	
	CDF_boys  toys_transitivity "Number of Violations of Transitivity in Toys Task" "0 15" "0(5)15"
	CDF_girls toys_transitivity "Number of Violations of Transitivity in Toys Task" "0 15" "0(5)15" "keep if toys_transitivity <= 15"	

	CDF_boys  sharing_transitivity "Number of Violations of Transitivity in Sharing Task" "0 8" "0(2)8"
	CDF_girls sharing_transitivity "Number of Violations of Transitivity in Sharing Task" "0 8" "0(2)8"	

	CDF_boys  risk_consistency "Number of Violations of Transitivity and/or Monotonicity in Risk Task" "0 8" "0(2)8"
	CDF_girls risk_consistency "Number of Violations of Transitivity and/or Monotonicity in Risk Task" "0 8" "0(2)8"

	CDF_boys  risk_dominated "Number of Dominated Choices in Risk Task" "0 8" "0(2)8"
	CDF_girls risk_dominated "Number of Dominated Choices in Risk Task" "0 8" "0(2)8"	
	
	
	************************************************************
	*** TABLE S18 - ITT on MEASURES of DECISION-MAKING QUALITY *
	************************************************************
	
	xtset cluster
	
	foreach var in toys_transitivity sharing_transitivity risk_consistency risk_dominated {
	
		preserve
		
			qui summ `var' if ITT == 0
			replace `var' = (`var' - r(mean))/r(sd)
		
			xtreg `var' male_ITT female_ITT male, r fe
			
		restore
	
	}
	
	qui tab cluster, gen(cluster_)
	
	clonevar TT = toys_transitivity 
	clonevar ST = sharing_transitivity 
	clonevar RC = risk_consistency 
	clonevar RD = risk_dominated
	
	rwolf TT ST RC RD, indepvar(female_ITT male_ITT) controls(cluster_* male) seed(70086298) reps(1000) 	
	
	******************************************************
	***	TABLE S21 - MARSHMALLOW Robustness Check 	   ***
	******************************************************	
	
	xtset cluster
	
	xtreg waiting_time male_ITT female_ITT male, r fe
	
		
	
	reg waiting_time ITT if male == 1, r
	
	tobit waiting_time ITT if male == 1, vce(r) ul(25)
	
	leebounds waiting_time ITT if male == 1, cie 	
	
	
	reg waiting_time ITT if male == 0, r
	
	tobit waiting_time ITT if male == 0, vce(r)  ul(25)
	
	leebounds waiting_time ITT if male == 0, cie 	
	
	
	*********************************************************
	***	FIG S6 - ITT on MEASURES of COGNITIVE FUNCTION *
	*********************************************************
	
	CDF_boys  working_memory "Working Memory"  "-4 4" "-4(2)4"
	CDF_girls working_memory "Working Memory"  "-4 4" "-4(2)4"

	CDF_boys  perceptual_reasoning "Perceptual Reasoning" "-4 4" "-4(2)4"
	CDF_girls perceptual_reasoning "Perceptual Reasoning" "-4 4" "-4(2)4"	

	CDF_boys  processing_speed "Processing Speed" "-4 4" "-4(2)4"
	CDF_girls processing_speed "Processing Speed" "-4 4" "-4(2)4"

	CDF_boys  verbal_comprehension "Verbal Comprehension" "-4 4" "-4(2)4"
	CDF_girls verbal_comprehension "Verbal Comprehension" "-4 4" "-4(2)4"	

	
	****************************************************************
	*** TABLE S22- INTENT-to_TREAT EFFECTS on COGNITIVE FUNCTION ***
	****************************************************************
	
	xtset cluster
	
	preserve
		
		foreach var in working_memory perceptual_reasoning verbal_comprehension processing_speed {
		
			qui summ `var' if ITT == 0
			replace `var' = (`var' - r(mean))/r(sd)
			
			xtreg `var' male_ITT female_ITT male, r fe
		
		}
	
	restore
	
	clonevar WM = working_memory 
	clonevar PR = perceptual_reasoning 
	clonevar PS = processing_speed 
	clonevar VC = verbal_comprehension
	
	
	rwolf WM PR VC PS, indepvar(female_ITT male_ITT) controls(cluster_* male) seed(70086298) reps(1000) 	
